Pensieve header: $\mathcal A$-calculus with q-commutation relations, $\xi x = - q x \xi$.

$\xi x = -q x \xi$

q𝒜-Calculus

In[∘]:=
```
WP[Wedge[u___], Wedge[v___]] :=
  Signature[{u, v}] * q^(Count[{u},ξ_] Count[{v},x_]) * Wedge @@ Sort[{u, v}];
WP[0, _] = WP[_, 0] = 0;
WP[A_, B_] :=
  Expand[Distribute[A ** B] /. (a_. * u_Wedge) ** (b_. * v_Wedge) :> a b WP[u, v]];
```

In[∘]:=
```
WExp[A_] := Module[{s = Wedge[], t = Wedge[], k = 0},
  While[t =!= 0, s += (t = Expand[WP[t, A] / (++k)])]; s]
```

In[∘]:= `WExp[x1 ∧ ξ1 + x2 ∧ ξ2 + x3 ∧ ξ3]`

Out[∘]=

$\text{Wedge}[\,] + x_1 \wedge \xi_1 + x_2 \wedge \xi_2 + x_3 \wedge \xi_3 - q\, x_1 \wedge x_2 \wedge \xi_1 \wedge \xi_2 - q\, x_1 \wedge x_3 \wedge \xi_1 \wedge \xi_3 - q\, x_2 \wedge x_3 \wedge \xi_2 \wedge \xi_3 - q^3\, x_1 \wedge x_2 \wedge x_3 \wedge \xi_1 \wedge \xi_2 \wedge \xi_3$

In[∘]:=
```
c_{X_,Ξ_}[w_Wedge] := Module[{i, ι},
   {i} = FirstPosition[w, X, {0}]; {ι} = FirstPosition[w, Ξ, {0}];
   Which[
    (i == 0) ∧ (ι == 0), w,
    (i > 0) ∧ (ι > 0), Times[
     (-1)^(i+ι), q^(Count[w[[;;i-1]],ξ_]-Count[w[[;;ι-1]],x_]), If[ι < i, 1, -q], Delete[w, {{i}, {ι}}]
    ],
    True, 0
   ]
  ];
c_{x_,ξ_}[ℰ_] := ℰ /. w_Wedge :> c_{x,ξ}[w]
```

In[∘]:=
```
WExp[x1 ∧ ξ1 + 2 x2 ∧ ξ2]
c_{x1,ξ2}@WExp[x1 ∧ ξ1 + 2 x2 ∧ ξ2]
```

Out[∘]=

$\text{Wedge}[\,] + x_1 \wedge \xi_1 + 2\, x_2 \wedge \xi_2 - 2\, q\, x_1 \wedge x_2 \wedge \xi_1 \wedge \xi_2$

Out[∘]=

$\text{Wedge}[\,] - 2\, x_2 \wedge \xi_1$

In[∘]:=

$$\text{WExp}[x_1\wedge\xi_1+x_2\wedge\xi_2+x_3\wedge\xi_3]$$

$$c_{x_1,\xi_3}\text{@WExp}[x_1\wedge\xi_1+x_2\wedge\xi_2+x_3\wedge\xi_3]$$

Out[∘]=

$$\text{Wedge}[\,]+x_1\wedge\xi_1+x_2\wedge\xi_2+x_3\wedge\xi_3-q\,x_1\wedge x_2\wedge\xi_1\wedge\xi_2-q\,x_1\wedge x_3\wedge\xi_1\wedge\xi_3-q\,x_2\wedge x_3\wedge\xi_2\wedge\xi_3-q^3\,x_1\wedge x_2\wedge x_3\wedge\xi_1\wedge\xi_2\wedge\xi_3$$

Out[∘]=

$$\text{Wedge}[\,]+x_2\wedge\xi_2-x_3\wedge\xi_1-q\,x_2\wedge x_3\wedge\xi_1\wedge\xi_2$$

In[∘]:= $\text{WExp}[x_2\wedge\xi_2-x_3\wedge\xi_1]$

Out[∘]=

$$\text{Wedge}[\,]+x_2\wedge\xi_2-x_3\wedge\xi_1-q\,x_2\wedge x_3\wedge\xi_1\wedge\xi_2$$



In[∘]:=

$$\mathcal{A}[X_{i\_,j\_,k\_,l\_}[S\_,T\_]]:=\mathcal{A}\Big[\{l,i\},\{j,k\},\langle|\xi_i\to S,\ x_j\to T,\ x_k\to S,\ \xi_l\to T|\rangle,$$
$$\text{Expand}\Big[T^{-1/2}\,\text{WExp}\Big[\text{Expand}\Big[\{\xi_l,\xi_i\}.\begin{pmatrix}1 & 1-T\\ 0 & T\end{pmatrix}.\{x_j,x_k\}\Big]\ /.\ \xi_{a\_}\,x_{b\_}\mapsto\xi_a\wedge x_b\Big]\Big]\Big];$$

In[∘]:= $\mathcal{A}[X_{1,2,3,4}[u,v]]$

Out[∘]=

$$\mathcal{A}\Big[\{4,1\},\{2,3\},\langle|\xi_1\to u,\ x_2\to v,\ x_3\to u,\ \xi_4\to v|\rangle,\ \frac{\text{Wedge}[\,]}{\sqrt{v}}-\frac{x_2\wedge\xi_4}{\sqrt{v}}-\sqrt{v}\ x_3\wedge\xi_1-\frac{x_3\wedge\xi_4}{\sqrt{v}}+\sqrt{v}\ x_3\wedge\xi_4+q\ \sqrt{v}\ x_2\wedge x_3\wedge\xi_1\wedge\xi_4\Big]$$

In[∘]:=

$$\mathcal{A}[X_{i\_,j\_,k\_,l\_}]:=\mathcal{A}[X_{i,j,k,l}[\tau_i,\tau_l]]$$

In[∘]:=

$$\mathcal{A}[\overline{X}_{i\_,j\_,k\_,l\_}[S\_,T\_]]:=\mathcal{A}\Big[\{i,j\},\{k,l\},\langle|\xi_i\to S,\ \xi_j\to T,\ x_k\to S,\ x_l\to T|\rangle,$$
$$\text{Expand}\Big[T^{1/2}\,\text{WExp}\Big[\text{Expand}\Big[\{\xi_i,\xi_j\}.\begin{pmatrix}T^{-1} & 0\\ 1-T^{-1} & 1\end{pmatrix}.\{x_k,x_l\}\Big]\ /.\ \xi_{a\_}\,x_{b\_}\mapsto\xi_a\wedge x_b\Big]\Big]\Big];$$
$$\mathcal{A}[\overline{X}_{i\_,j\_,k\_,l\_}]:=\mathcal{A}[\overline{X}_{i,j,k,l}[\tau_i,\tau_j]];$$

In[∘]:=

$$\mathcal{A}[P_{i\_,j\_}[T\_]]:=\mathcal{A}[\{i\},\{j\},\langle|\xi_i\to T,\ x_j\to T|\rangle,\ \text{WExp}[\xi_i\wedge x_j]];$$
$$\mathcal{A}[P_{i\_,j\_}]:=\mathcal{A}[P_{i,j}[\tau_i]]$$

In[∘]:=

```
𝒜 /: α_ × 𝒜[is_, os_, cs_, w_] := 𝒜[is, os, cs, Expand[α w]]
𝒜 /: 𝒜[is1_, os1_, cs1_, w1_] + 𝒜[is2_, os2_, cs2_, w2_] /;
   (Sort@is1 == Sort@is2) ∧ (Sort@os1 == Sort@os2) ∧
    (Sort@Normal@cs1 == Sort@Normal@cs2) := 𝒜[is1, os1, cs1, w1 + w2]
```

In[∘]:=
```
𝒜 /: 𝒜[is1_, os1_, _, w1_] ≡ 𝒜[is2_, os2_, _, w2_] :=
  TrueQ[(Sort@is1 === Sort@is2) ∧ (Sort@os1 === Sort@os2) ∧ PowerExpand[w1 == w2]]
```

In[∘]:=
```
𝒜 /: 𝒜[is1_, os1_, cs1_, w1_] 𝒜[is2_, os2_, cs2_, w2_] :=
  𝒜[is1 ⋃ is2, os1 ⋃ os2, Join[cs1, cs2], WP[w1, w2]]
```

In[∘]:=
```
Short[𝒜[X_{2,4,3,1}[S, T]] 𝒜[X̄_{3,4,6,5}], 5]
```

Out[∘]//Short=

$$\mathcal{A}\Big[\{1,2,3,4\},\{3,4,5,6\},\langle|\xi_2\to S,x_4\to T,x_3\to S,\xi_1\to T,\xi_3\to\tau_3,\xi_4\to\tau_4,x_6\to\tau_3,x_5\to\tau_4|\rangle,$$
$$\frac{\sqrt{\tau_4}\,\text{Wedge}[\,]}{\sqrt{T}}-\frac{\sqrt{\tau_4}\,x_3\wedge\xi_1}{\sqrt{T}}+\sqrt{T}\sqrt{\tau_4}\,x_3\wedge\xi_1-\sqrt{T}\sqrt{\tau_4}\,x_3\wedge\xi_2-\frac{\sqrt{\tau_4}\,x_4\wedge\xi_1}{\sqrt{T}}-\frac{\sqrt{\tau_4}\,x_5\wedge\xi_4}{\sqrt{T}}-$$
$$\frac{x_6\wedge\xi_3}{\sqrt{T}\sqrt{\tau_4}}+\ll 35\gg+\frac{q^3\sqrt{T}\,x_3\wedge x_4\wedge x_6\wedge\xi_1\wedge\xi_2\wedge\xi_4}{\sqrt{\tau_4}}-q^3\sqrt{T}\sqrt{\tau_4}\,x_3\wedge x_4\wedge x_6\wedge\xi_1\wedge\xi_2\wedge\xi_4-$$
$$\frac{q^3\,x_3\wedge x_5\wedge x_6\wedge\xi_1\wedge\xi_3\wedge\xi_4}{\sqrt{T}\sqrt{\tau_4}}+\frac{q^3\sqrt{T}\,x_3\wedge x_5\wedge x_6\wedge\xi_1\wedge\xi_3\wedge\xi_4}{\sqrt{\tau_4}}-\frac{q^3\sqrt{T}\,x_3\wedge x_5\wedge x_6\wedge\xi_2\wedge\xi_3\wedge\xi_4}{\sqrt{\tau_4}}-$$
$$\frac{q^3\,x_4\wedge x_5\wedge x_6\wedge\xi_1\wedge\xi_3\wedge\xi_4}{\sqrt{T}\sqrt{\tau_4}}+\frac{q^6\sqrt{T}\,x_3\wedge x_4\wedge x_5\wedge x_6\wedge\xi_1\wedge\xi_2\wedge\xi_3\wedge\xi_4}{\sqrt{\tau_4}}\Big]$$

In[∘]:=
```
c_{h_,t_}@𝒜[is_, os_, cs_, w_] := 𝒜[
    DeleteCases[is, t], DeleteCases[os, h], KeyDrop[cs, {x_h, ξ_t}], c_{x_h,ξ_t}[w]
  ] /. If[MatchQ[cs[ξ_t], τ_], cs[ξ_t] → cs[x_h], cs[x_h] → cs[ξ_t]];
```

In[∘]:=
```
c_{4,4}[𝒜[X_{2,4,3,1}[S, T]] 𝒜[X̄_{3,4,6,5}]]
```

Out[∘]=

$$\mathcal{A}\Big[\{1,2,3\},\{3,5,6\},\langle|\xi_2\to S,x_3\to S,\xi_1\to T,\xi_3\to\tau_3,x_6\to\tau_3,x_5\to T|\rangle,$$
$$\text{Wedge}[\,]-x_3\wedge\xi_1+T\,x_3\wedge\xi_1-T\,x_3\wedge\xi_2-x_5\wedge\xi_1-x_6\wedge\xi_1+\frac{x_6\wedge\xi_1}{T}-\frac{x_6\wedge\xi_3}{T}+$$
$$q\,T\,x_3\wedge x_5\wedge\xi_1\wedge\xi_2-q\,x_3\wedge x_6\wedge\xi_1\wedge\xi_2+q\,T\,x_3\wedge x_6\wedge\xi_1\wedge\xi_2+q\,x_3\wedge x_6\wedge\xi_1\wedge\xi_3-$$
$$\frac{q\,x_3\wedge x_6\wedge\xi_1\wedge\xi_3}{T}-q\,x_3\wedge x_6\wedge\xi_2\wedge\xi_3-\frac{q\,x_5\wedge x_6\wedge\xi_1\wedge\xi_3}{T}-q^3\,x_3\wedge x_5\wedge x_6\wedge\xi_1\wedge\xi_2\wedge\xi_3\Big]$$

In[∘]:=
```
c@𝒜[is_, os_, cs_, w_] := Fold[c_{#2,#2}[#1] &, 𝒜[is, os, cs, w], is ⋂ os]
𝒜[{A_𝒜}] := c[A];
𝒜[{A1_𝒜, As__𝒜}] := Module[{A2},
  A2 = First@MaximalBy[{As}, Length[A1⟦1⟧ ⋂ #⟦2⟧] + Length[A1⟦2⟧ ⋂ #⟦1⟧] &];
  𝒜[Join[{c[A1 A2]}, DeleteCases[{As}, A2]]] ]
𝒜[𝒪s_List] := 𝒜[𝒜 /@ 𝒪s]
```

In[∘]:=
```
c[𝒜[X_{2,4,3,1}[S, T]] 𝒜[X̄_{3,4,6,5}]]
```

Out[∘]=

$$\mathcal{A}[\{1,2\},\{5,6\},\langle|\xi_2\to S,\xi_1\to T,x_6\to S,x_5\to T|\rangle,\text{Wedge}[\,]-x_5\wedge\xi_1-x_6\wedge\xi_2-q\,x_5\wedge x_6\wedge\xi_1\wedge\xi_2]$$

In[∘]:=
```
𝒜@{𝒜[X_{2,4,3,1}[S, T]], 𝒜[X̄_{3,4,6,5}]}
```
Out[∘]=
$$\mathcal{A}[\{1,2\},\{5,6\},\langle| \xi_2\to S,\ \xi_1\to T,\ x_6\to S,\ x_5\to T |\rangle,\ \text{Wedge}[\,]-x_5\wedge\xi_1-x_6\wedge\xi_2-q\,x_5\wedge x_6\wedge\xi_1\wedge\xi_2]$$

Skein Relations



In[∘]:=
```
𝒜@{X̄_{4,1,6,3}[v, u], X̄_{3,2,5,4}}
```
Out[∘]=
$$\mathcal{A}\Big[\{1,2\},\{5,6\},\langle| \xi_2\to v,\ x_5\to u,\ \xi_1\to u,\ x_6\to v |\rangle,$$
$$\sqrt{u}\,\sqrt{v}\,\text{Wedge}[\,]-\frac{\sqrt{u}\,x_5\wedge\xi_1}{\sqrt{v}}+\frac{\sqrt{u}\,x_5\wedge\xi_2}{\sqrt{v}}-\sqrt{u}\,\sqrt{v}\,x_5\wedge\xi_2+\frac{\sqrt{v}\,x_6\wedge\xi_1}{\sqrt{u}}-\sqrt{u}\,\sqrt{v}\,x_6\wedge\xi_1-$$
$$\frac{\sqrt{v}\,x_6\wedge\xi_2}{\sqrt{u}}-\frac{q\,\sqrt{u}\,x_5\wedge x_6\wedge\xi_1\wedge\xi_2}{\sqrt{v}}-\frac{q\,\sqrt{v}\,x_5\wedge x_6\wedge\xi_1\wedge\xi_2}{\sqrt{u}}+q\,\sqrt{u}\,\sqrt{v}\,x_5\wedge x_6\wedge\xi_1\wedge\xi_2\Big]$$

Reidemeister 2



In[∘]:=
```
𝒜@{X_{2,4,3,1}[S, T], X̄_{3,4,6,5}} ≡ 𝒜@{P_{1,5}[T], P_{2,6}[S]}
```
Out[∘]=
```
True
```

In[∘]:=
```
𝒜@{X̄_{3,1,2,4}[S, T], X_{6,5,3,4}} ≡ 𝒜@{P_{1,5}[T], P_{6,2}[S]}
```
Out[∘]=
```
True
```

Reidemeister 3



In[∘]:= $\mathcal{A}$@{$X_{2,5,4,1}$[T_2, T_1], $X_{3,7,6,5}$[T_3, T_1], $X_{6,9,8,4}$} ≡ $\mathcal{A}$@{$X_{3,5,4,2}$[T_3, T_2], $X_{4,6,8,1}$[T_3, T_1], $X_{5,7,9,6}$}

Out[∘]=

True

Reidemeister 1



In[∘]:= {$\mathcal{A}$@{$X_{3,3,2,1}$} ≡ $\tau_1^{-1/2}$ $\mathcal{A}$@{$P_{1,2}$}, $\mathcal{A}$@{$X_{1,2,3,3}$} ≡ $\tau_1^{1/2}$ $\mathcal{A}$@{$P_{1,2}$},
$\mathcal{A}$@{$\overline{X}_{1,3,3,2}$} ≡ $\tau_1^{-1/2}$ $\mathcal{A}$@{$P_{1,2}$}, $\mathcal{A}$@{$\overline{X}_{3,1,2,3}$} ≡ $\tau_1^{1/2}$ $\mathcal{A}$@{$P_{1,2}$}}

Out[∘]=

{True, True, True, True}

The Relation with the Multivariable Alexander Polynomial



In[∘]:= MVA = $u^{-1/2}$ $v^{-1/2}$ $w^{-1/2}$ (u - 1) (v - 1) (w - 1);

In[∘]:= A = {$\overline{X}_{1,12,2,13}$[u, v], $\overline{X}_{13,2,6,3}$, $X_{8,4,9,3}$, $X_{4,10,5,9}$, $X_{6,17,7,16}$[v, w],
$X_{15,8,16,7}$, $\overline{X}_{14,10,15,11}$, $\overline{X}_{11,17,12,14}$} // $\mathcal{A}$ // Last // Factor

Out[∘]=

$$\frac{(-1+u)^2\,(-1+v)\,(-1+w)\,(\text{Wedge}[\,]-x_5\wedge\xi_1)}{u\,v}$$

In[∘]:= A == $u^{-1/2}$ (u - 1) u^0 $v^{-1/2}$ $w^{1/2}$ MVA (Wedge[] - $x_5 \wedge \xi_1$)

Out[∘]=

True

Overcrossings Commute but Undercrossings don’t



In[∘]:= $\mathcal{A}$@$\{X_{2,7,5,1}, X_{3,4,6,7}\}$ ≡ $\mathcal{A}$@$\{X_{3,7,6,1}, X_{2,4,5,7}\}$

Out[∘]=

True

In[∘]:= $\mathcal{A}$@$\{\overline{X}_{1,2,7,5}, \overline{X}_{7,3,4,6}\}$ ≡ $\mathcal{A}$@$\{\overline{X}_{1,3,7,6}, \overline{X}_{7,2,4,5}\}$

Out[∘]=

False

The Conway Relation



In[∘]:= $\mathcal{A}$@$\{X_{2,3,4,1}[T, T]\}$ - $\mathcal{A}$@$\{\overline{X}_{1,2,3,4}[T, T]\}$ ≡ $(T^{-1/2} - T^{1/2})$ $\mathcal{A}$@$\{P_{1,4}[T], P_{2,3}[T]\}$

Out[∘]=

True

Conway’s Second Set of Identities



In[∘]:= $\mathcal{A}$@$\{X_{2,4,3,1}[v, u], X_{4,6,5,3}\}$ + $\mathcal{A}$@$\{\overline{X}_{1,2,4,3}[u, v], \overline{X}_{3,4,6,5}\}$ ≡
$(u^{1/2} v^{1/2} + u^{-1/2} v^{-1/2})$ $\mathcal{A}$@$\{P_{1,5}[u], P_{2,6}[v]\}$

Out[∘]=

True

In[∘]:= $\mathcal{A}$@$\{\overline{X}_{4,1,6,3}[v, u], \overline{X}_{3,2,5,4}\}$ + $\mathcal{A}$@$\{X_{1,6,3,4}[u, v], X_{2,5,4,3}\}$ ≡
$(u^{1/2} v^{-1/2} + u^{-1/2} v^{1/2})$ $\mathcal{A}$@$\{P_{1,5}[u], P_{2,6}[v]\}$

Out[∘]=

True



In[∘]:= $\mathcal{A}@\{X_{2,3,4,1}\} + \mathcal{A}@\{\overline{X}_{2,1,4,3}\} \equiv \left(\tau_1^{1/2} + \tau_1^{-1/2}\right) \mathcal{A}@\{P_{1,3}, P_{2,4}\}$

Out[∘]=

True

In[∘]:= $\mathcal{A}@\{\overline{X}_{1,2,3,4}\} + \mathcal{A}@\{X_{1,4,3,2}\} \equiv \left(\tau_2^{1/2} + \tau_2^{-1/2}\right) \mathcal{A}@\{P_{1,3}, P_{2,4}\}$

Out[∘]=

True

Conway’s Third Identity



In[∘]:= $\mathcal{A}@\left\{X_{6,4,9,1}, \overline{X}_{4,5,7,8}, \overline{X}_{2,3,5,6}\right\} + \mathcal{A}@\left\{X_{2,4,5,1}, \overline{X}_{4,3,7,6}, X_{6,8,9,5}\right\} \equiv$
$\mathcal{A}@\left\{\overline{X}_{1,6,4,9}, X_{5,7,8,4}, X_{3,5,6,2}\right\} + \mathcal{A}@\left\{\overline{X}_{1,2,4,5}, X_{3,7,6,4}, \overline{X}_{5,6,8,9}\right\}$

Out[∘]=

True

Virtual version



In[∘]:= $\mathcal{A}@\left\{X_{3,7,6,1}, \overline{X}_{7,2,4,5}\right\} + \mathcal{A}@\{X_{2,4,7,1}, X_{3,5,6,7}\} \equiv \mathcal{A}@\{X_{3,7,6,2}, X_{7,4,5,1}\} + \mathcal{A}@\left\{\overline{X}_{1,2,7,5}, X_{3,4,6,7}\right\}$

Out[∘]=

True

Jun Murakami’s Fifth Axiom



In[∘]:= $\mathcal{A}$@{$X_{1,4,2,5}$[T, S], $X_{4,3,5,2}$} ≡ $\frac{\sqrt{S}\ (1-T)}{\sqrt{T}}$ $\mathcal{A}$@{$P_{1,3}$[T]}

Out[∘]=

True



Virtual versions

In[∘]:= $\mathcal{A}$@{$X_{3,2,3,1}$[S, T]} ≡ $\left(T^{-1/2} - T^{1/2}\right)$ $\mathcal{A}$@{$P_{1,2}$[T]}

Out[∘]=

True

In[∘]:= $\mathcal{A}$@{$X_{1,3,2,3}$}

Out[∘]=

$\mathcal{A}$[{1}, {2}, ⟨| $\xi_1 \to \tau_1$, $x_2 \to \tau_1$ |⟩, 0]

Jun Murakami's Third Axiom



In[∘]:= $\mathcal{A}_{2112}$ = $\mathcal{A}$@{$X_{3,8,7,2}$, $X_{7,10,9,1}$, $X_{10,11,4,9}$, $X_{8,6,5,11}$};
$\mathcal{A}_{1221}$ = $\mathcal{A}$@{$X_{2,8,7,1}$, $X_{3,10,9,8}$, $X_{10,6,11,9}$, $X_{11,5,4,7}$};
$\mathcal{A}_{2211}$ = $\mathcal{A}$@{$X_{3,8,7,2}$, $X_{8,6,9,7}$, $X_{9,11,10,1}$, $X_{11,5,4,10}$};
$\mathcal{A}_{1122}$ = $\mathcal{A}$@{$X_{2,8,7,1}$, $X_{8,9,4,7}$, $X_{3,11,10,9}$, $X_{11,6,5,10}$};
$\mathcal{A}_{11}$ = $\mathcal{A}$@{$X_{2,8,7,1}$, $X_{8,5,4,7}$, $P_{3,6}$}; $\mathcal{A}_{22}$ = $\mathcal{A}$@{$X_{3,8,7,2}$, $X_{8,6,5,7}$, $P_{1,4}$};
$\mathcal{A}_{\phi}$ = $\mathcal{A}$@{$P_{1,4}$, $P_{2,5}$, $P_{3,6}$};
g_+[z_] := $z^{1/2} + z^{-1/2}$; g_-[z_] := $z^{1/2} - z^{-1/2}$;
$g_+[\tau_1]\ g_-[\tau_2]\ \mathcal{A}_{2112} - g_-[\tau_2]\ g_+[\tau_3]\ \mathcal{A}_{1221} - g_-[\tau_3/\tau_1]\ (\mathcal{A}_{2211} + \mathcal{A}_{1122})$ +
$g_-[\tau_2\,\tau_3/\tau_1]\ g_+[\tau_3]\ \mathcal{A}_{11} - g_+[\tau_1]\ g_-[\tau_1\,\tau_2/\tau_3]\ \mathcal{A}_{22}$ ≡ $g_-\left[\tau_3^2 / \tau_1^2\right]\ \mathcal{A}_{\phi}$

Out[∘]=

True

The Naik-Stanford Double Delta Move



In[∘]:= Timing[𝒜@{$X_{6,10,28,24}$[w, v], $\overline{X}_{28,3,29,19}$[w, v], $X_{26,20,27,19}$[w, v], $\overline{X}_{27,23,11,24}$[w, v],
$X_{1,12,13,30}$[u, w], $\overline{X}_{13,5,14,25}$[u, w], $X_{17,26,18,25}$[u, w], $\overline{X}_{18,29,8,30}$[u, w],
$X_{4,7,22,15}$[v, u], $\overline{X}_{22,2,23,16}$[v, u], $X_{20,17,21,16}$[v, u], $\overline{X}_{21,14,9,15}$[v, u]} ≡
𝒜@{$X_{5,9,25,21}$[w, v], $\overline{X}_{25,4,26,22}$[w, v], $X_{29,23,30,22}$[w, v], $\overline{X}_{30,20,12,21}$[w, v],
$X_{2,11,16,27}$[u, w], $\overline{X}_{16,6,17,28}$[u, w], $X_{14,29,15,28}$[u, w], $\overline{X}_{15,26,7,27}$[u, w],
$X_{3,8,19,18}$[v, u], $\overline{X}_{19,1,20,13}$[v, u], $X_{23,14,24,13}$[v, u], $\overline{X}_{24,17,10,18}$[v, u]}]

Out[∘]=

{182.188, True}

Virtual Version 1



In[∘]:= 𝒜@{$X_{1,8,11,3}$[u, v], $\overline{X}_{11,2,12,7}$[u, v], $X_{12,10,13,4}$[u, w], $\overline{X}_{13,5,6,9}$[u, w]} ≡
𝒜@{$X_{1,10,11,4}$[u, w], $\overline{X}_{11,5,12,9}$[u, w], $X_{12,8,13,3}$[u, v], $\overline{X}_{13,2,6,7}$[u, v]}

Out[∘]=

True

Virtual Version 2



In[∘]:= $\mathcal{A}$@$\{\overline{X}_{20,1,10,13}[v,u], X_{3,14,19,13}[v,u], X_{14,11,15,21}[u,w], \overline{X}_{15,6,7,22}[u,w], X_{2,12,16,22}[u,w], \overline{X}_{16,5,17,21}[u,w], \overline{X}_{19,17,9,18}[v,u], X_{4,8,20,18}[v,u]\}$ ≡ $\mathcal{A}$@$\{X_{1,11,13,21}[u,w], \overline{X}_{13,6,14,22}[u,w], \overline{X}_{20,14,10,15}[v,u], X_{3,7,19,15}[v,u], \overline{X}_{19,2,9,16}[v,u], X_{4,17,20,16}[v,u], X_{17,12,18,22}[u,w], \overline{X}_{18,5,8,21}[u,w]\}$

Out[∘]=

True